
Basic Data Operation

This is a basic data operation. It represents a fundamental action on a basic data type.
You may see other such operations in the Basic Data Operations category, or:


Integer Operations
Arithmetic |
Comparison
Boolean Operations
Bitwise |
Logical
String Operations
Concatenation |
Interpolation |
Comparison |
Matching
Memory Operations
Pointers & references |
Addresses


Get two integers from the user.

Then,   display a message if the first integer is:

   less than, 
   equal to,   or 
   greater than
the second integer.



Test the condition   for each case separately,   so that   all three comparison operators are used   in the code.



Related task

   String comparison


/*REXX program  prompts  for  two integers,   compares them,  and  displays the results.*/
numeric digits 2000                              /*for the users that really go ka─razy.*/
@=copies('─', 20)                                /*eyeball catcher for the user's eyen. */
a=getInt(@  'Please enter your 1st integer:')    /*obtain the 1st integer from the user.*/
b=getInt(@  'Please enter your 2nd integer:')    /*   "    "  2nd    "      "   "    "  */
say
      if a<b  then say  @   a    ' is less than '        b
      if a=b  then say  @   a    ' is equal to '         b
      if a>b  then say  @   a    ' is greater than '     b
exit                                             /*stick a fork in it,  we're all done. */
/*──────────────────────────────────────────────────────────────────────────────────────*/
getInt:       do forever;     say                /*keep prompting the user until success*/
                              say arg(1)         /*display the prompt message to console*/
              parse pull x                       /*obtain  X,  and keep its case intact.*/
                 select
                 when x=''               then call serr "No argument was entered."
                 when words(x)>1         then call serr 'Too many arguments entered.'  x
                 when \datatype(x, 'N')  then call serr "Argument isn't numeric:"      x
                 when \datatype(x, 'W')  then call serr "Argument isn't an integer:"   x
                 otherwise    return x           /* [↑]  Eureka!   Return # to invoker. */
                 end   /*select*/
              end      /*forever*/
/*──────────────────────────────────────────────────────────────────────────────────────*/
serr:  say @  '***error*** '    arg(1);        say @  "Please try again.";          return